Уважаемый читатель!

Я очень рад за тебя, что ты взял в руки эту рукопись. Когда мы с сыном начинали разбираться с языком программирования Small Basic у нас ее не было под рукой, а нам очень не хватало такого справочника, в котором были бы отображены все команды языка и краткие примеры для каждой из них.

Теперь у тебя есть эта рукопись, которая ещё не закончена и мне как её составителю хотелось бы, что бы ты уважаемый читатель принял бы активное участие. Ты спросишь – «как, ведь я ещё так мало знаю?» Ну и что, читая её у тебя, наверняка возникнут вопросы, так что можешь, смело их задавать мне или любому пользователю форума http://forum.smallbasic.ru/. А пожелания по содержанию присылать мне в личку: Papa_Alex_2000. Я думаю, совместными усилиями мы превратим эту рукопись в настоящую настольную книгу.

Удачи всем нам!

С уважением, Papa_Alex_2000

P.S. Все материалы представленные в этой рукописи взяты из открытых источников интернета, с ними можно делать что угодно, но помни о тех кто это собрал, перевел и дал возможность тебе пользоваться.


3Синтаксис языка Small Basic


6Array (Работа с массивом данных)


10Clock (Работа с ситемными часами)


11Desktop (Работа с рабочим столом)


12Dictionary (Работа с интернет-словарем)


13File (Работа с внешним файлом)


17Flickr (Доступ к Интернет-сервису фотографий)


18GraphicsWindow (Работа в графическом окне)


25ImageList (Работа с внешним изображением)


26Math (Работа с математическими операторами)


30Mouse (Работа со свойствами мыши)


32Network (Работа с сетью)


33Program (Контроль над выполнением программы)


34Shapes (Работа с фигурами)


38Sound (Работа со звуком)


40Stack (Работа со стеком)


42Text (Работа с текстом)


45TextWindow ( Работа с текстовым окном)


48Timer (Работа с повтором действий)


49Turtle (Работа с черепашкой)


52Ключевые слова





Синтаксис языка Small Basic

Шпаргалка как создать и получить доступ к основным операциям в Small Basic. Её можно использовать, когда нужно вспомнить синтаксис, но не хочется смотреть руководство пользователя.

СОЗДАТЬ:
Array (Массив):
myArray(мой массив)[1] = "myString(моя запись)"

TextWindow.WriteLine(Array.GetItemCount(myArray))
Directory (Директорию):
File.CreateDirectory("C:\temp\myTempDir(Моя временная директория)")

Ellipse (Эллипс):
GraphicsWindow.DrawEllipse(10, 15, 75, 25)
' или...

myEllipse(мой эллипс) = Shapes.AddEllipse(100, 50)

Flickr image:
GraphicsWindow.DrawImage(Flickr.GetPictureOfMoment(), 0, 0)
Оператор For:

For i = 1 To 10
    TextWindow.WriteLine(i)

EndFor
Оператор If/Then:

If Clock.Year = 2010 Then

    TextWindow.WriteLine("2010")
Else

    TextWindow.WriteLine("Не 2010")

EndIf

Image(Изображение):

myImage(мое изображение) = Shapes.AddImage(ImageList.LoadImage("http://www.microsoft.com/about/images/home_blue.jpg"))

Label(Метка):

Goto myLabel(моя Метка)
myLabel(моя Метка):

    TextWindow.WriteLine("Привет, Мир!")

Line(Линия):

GraphicsWindow.DrawLine(10, 20, 100, 50)
' или...

myLine(моя линия) = Shapes.AddLine(15, 20, 40, 65)

Rectangle(Прямоугольник):
GraphicsWindow.DrawRectangle(5, 15, 80, 35)

' или...

myRectangle(мой прямоугольник) = Shapes.AddRectangle(50, 35)

Shape(Фигура):

myEllipse(мой эллипс) = Shapes.AddEllipse(100, 50)

Stack(Стек):

Stack.PushValue("myStack(мой стек)", "myValue(мое значение)")
TextWindow.WriteLine(Stack.PopValue("myStack(мой стек)"))

Subroutine(Процедура):

mySub()
Sub mySub

    TextWindow.WriteLine("Привет, Мир!")

EndSub

Triangle(Треугольник):

GraphicsWindow.DrawTriangle(10, 15, 50, 100, 75, 25)
' или...

myTriangle(мой треугольник) = Shapes.AddTriangle(30, 10, 100, 75, 145, 35)

ПОЛУЧЕНИЕ ДОСТУПА:
Clock functions: 

TextWindow.WriteLine(Clock.Date)
Desktop:

TextWindow.WriteLine(Desktop.Height)
Dictionary:

TextWindow.WriteLine(Dictionary.GetDefinition("banana"))
File functions:

TextWindow.WriteLine(File.GetTemporaryFilePath())
Graphics window:

GraphicsWindow.DrawBoundText(20, 10, 80, "Привет, Мир!")
Image:

GraphicsWindow.DrawImage(ImageList.LoadImage("http://www.microsoft.com/about/images/home_blue.jpg"), 0, 0)
Image list:

TextWindow.WriteLine(ImageList.GetHeightOfImage(ImageList.LoadImage("http://www.microsoft.com/about/images/home_blue.jpg")))
Math functions:

TextWindow.WriteLine(Math.SquareRoot(25))
Mouse:

GraphicsWindow.MouseDown = OnMouseDown
Sub OnMouseDown
  If Mouse.IsLeftButtonDown Then
    GraphicsWindow.Title = "Левая кнопка мыши нажата"
  Else
    GraphicsWindow.Title = "Левая кнопка мыши не нажата"
  EndIf
EndSub
Network functions:

TextWindow.WriteLine(Network.GetWebPageContents("http://www.microsoft.com/"))
Program functions:

TextWindow.WriteLine(Program.Directory)
Sound functions:

Sound.PlayBellRingAndWait()
Stack:

Stack.PushValue("myStack(мой стек)", "myValue(моё значение)")
TextWindow.WriteLine(Stack.PopValue("myStack(мой стек"))
Text functions:

TextWindow.WriteLine(Text.GetLength("myText(мой текст)"))
Text window:

TextWindow.WriteLine("Привет, Мир!")
Timer:

Timer.Tick = OnTick
Timer.Interval = 5000
Sub OnTick
  TextWindow.WriteLine(Clock.Time)
EndSub
Turtle:

Turtle.Move(100) 

Array (Работа с массивом данных)
Массив — это пронумерованная последовательность величин одинакового типа, обозначаемая одним именем. Элементы массива располагаются в последовательных ячейках памяти, обозначаются именем массива и индексом. Каждое из значений, составляющих массив, называется его компонентой (или элементом массива).

Массив данных в программе рассматривается как переменная структурированного типа. Массиву присваивается имя, посредством которого можно ссылаться как на массив данных в целом, так и на любую из его компонент.

Вообще, массив – однородный, упорядоченный структурированный тип данных с прямым доступом к элементам.

Переменные, представляющие компоненты массивов, называются переменными с индексами в отличие от простых переменных, представляющих в программе элементарные данные. Индекс в обозначении компонент массивов может быть константой, переменной или выражением порядкового типа (целочисленный, логический, символьный, перечислимый, диапазон).

Если за каждым элементом массива закреплен только один его порядковый номер, то такой массив называется линейным. Вообще количество индексов элементов массива определяет размерность массива. По этом признаку массивы делятся на одномерные (линейные), двумерные, трёхмерные и т.д.

Пример: числовая последовательность четных натуральных чисел 2, 4, 6, ..., N представляет собой линейный массив, элементы которого можно обозначить А[1]=2, А[2]=4, А[3]=6, ..., А[К]=2*(К+1), где К — номер элемента, а 2, 4, 6, ..., N — значения. Индекс (порядковый номер элемента) записывается в квадратных скобках после имени массива.

Например, A[7] — седьмой элемент массива А; D[6] — шестой элемент массива D.

Для размещения массива в памяти компьютера отводится поле памяти, размер которого определяется типом, длиной и количеством компонент массива. 

Операции
ContainsIndex

Array.ContainsIndex(arrayName, index) - Проверяет, содержит ли массив указанный индекс. Это очень полезно, при принятии решения, был ли индекс массива инициализирован каким-либо значением или нет.
arrayName - Массив для проверки. 
Index - Индекс для проверки.
Возвращает -  "True" или "False" в зависимости от того, существовал ли индекс в указанном массиве.
ContainsValue
Array.ContainsValue(arrayName, value) - Проверяет, содержит ли массив указанное значение. Это очень полезно, при принятии решения, было ли значение массива размещено в каком-либо индексе.
arrayName - Массив для проверки.
Value - Значение для проверки.
Возвращает - "True" или "False" в зависимости от того существовал ли индекс в указанном массиве.

GetAllIndices

Array.GetAllIndices(arrayName) - Возвращает все индексы для массива в виде другого массива.

arrayName - Массив, индексы которого надо получить.
Возвращает - Массив, который содержит индексы для указанного массива. Индекс возвращаемого массива начинается с 1.
GetItemCount
Array.GetItemCount(arrayName) - Возвращяет количество элементов в массиве.
arrayName - Массив для подсчета количества элементов.
Возвращает - Число элементов в заданном массиве.
IsArray
Array.IsArray(array) - Провереяет, является ли указанная переменная массивом.

arrayName - Переменная для проверки.
Возвращает - "True", если указанная переменная является массивом. В противном случае - "False".
GetValue

Array.GetValue(arrayName, index) - Получает значение для данного массива и индекса.
arrayName – Имя массива из которого необходимо получить значение.
Index - Индекс для которого необходимо получить значение.
Возвращает – Значение заданного индекса в указанном массиве.
RemoveValue

Array.RemoveValue(arrayName, index) - Удаляет элемент массива по указанному индексу.
arrayName – Имя массив из которого необходимо удалить значение.
Index - Индекс для которого необходимо удалить значение.
Возвращает - Ничего
SetValue

Array.SetValue(arrayName, index, value) - Устанавливает значение для данного массива и индекса.
arrayName – Имя массива в который необходимо добавить значение.
Index - Индекс для которого необходимо добавить значение.
Value – Добавляемое значение.
Возвращает – Ничего
	Microsoft Small Basic поддерживает массивы, которые позволяют хранить значения в оперативной память компьютера. В отличие от стека, который позволяет добавлять, читать и удалять элементы в определенном порядке, массивы позволяют добавлять, читать и удалять элементы в произвольном порядке. Кроме того, стек позволяет работать только с одномерным массивом, а массивы позволяют с многомерными величинами. Следующий пример иллюстрирует работу с массивами:

' Добавление продуктов к массиву food.
food[1] = "Яблоко"
food[2] = "Банан"
food[3] = "Кокос"
' Какие продукты находится в массиве food?
For i = 1 To Array.GetItemCount(food)
  TextWindow.WriteLine("food[" + i + "] содержит " + food[i])
EndFor
TextWindow.WriteLine("-----")
TextWindow.WriteLine("food это массив: " + Array.IsArray(food))
' Массив фруктов fruit все же существует?
TextWindow.WriteLine("fruit это массив: " + Array.IsArray(fruit))
TextWindow.WriteLine("food[2] существует: " + Array.ContainsIndex(food, 2))
TextWindow.WriteLine("food содержит Кокос: " + Array.ContainsValue(food, "Кокос"))
TextWindow.WriteLine("food содержит Виноград: " + Array.ContainsValue(food, "Виноград"))
TextWindow.WriteLine("food содержит " + Array.GetItemCount(food) + " предмета(ов).")
' Copy indices (not values) to fruit array. Копирует индексы (не значения) fruit массивa.
fruit = Array.GetAllIndices(food)
TextWindow.WriteLine("-----")
' Что в массиве fruit?
WhatIsInFruit()
TextWindow.WriteLine("-----")
' Обратите внимание, что значения не скопированы в массив fruit.
TextWindow.WriteLine("fruit is an array now: " + Array.IsArray(fruit))
TextWindow.WriteLine("fruit[2] exists: " + Array.ContainsIndex(fruit, 2))
TextWindow.WriteLine("fruit содержит Банан: " + Array.ContainsValue(fruit, "Банан"))
TextWindow.WriteLine("fruit содержит Виноград: " + Array.ContainsValue(fruit, "Виноград"))
TextWindow.WriteLine("fruit содержит " + Array.GetItemCount(fruit) + " предмета(ов).")
fruit[1] = "Яблоко"
fruit[2] = "Банан"
fruit[3] = "Кокос"
fruit[4] = "Виноград"
TextWindow.WriteLine("-----")
' Что сейчас в массиве fruit?
TextWindow.WriteLine("fruit содержит сейчас Банан: " + Array.ContainsValue(fruit, "Банан"))
TextWindow.WriteLine("fruit содержит сейчас Виноград: " + Array.ContainsValue(fruit, "Виноград"))
TextWindow.WriteLine("fruit сейчас содержит " + Array.GetItemCount(fruit) + " предмета(ов).")
TextWindow.WriteLine("-----")
WhatIsInFruit()  

Sub WhatIsInFruit
  For i = 1 To Array.GetItemCount(fruit)
    TextWindow.WriteLine("fruit[" + i + "] сейчас содержит " + fruit[i])
  EndFor 

EndSub
' Двумерный массив.
fruitSalad[1][1] = "Яблоко"
fruitSalad[1][2] = "Банан"
fruitSalad[1][3] = "Кокос"
fruitSalad[2][1] = "Виноград"
fruitSalad[2][2] = "Апельсин"
fruitSalad[2][3] = "Персик"
TextWindow.WriteLine("-----")
For i = 1 To 2
  For j = 1 To 3
    TextWindow.WriteLine("fruitSalad[" + i + "][" + j + "] содержит " + fruitSalad[i][j])
  EndFor
EndFor
mixedSalad["fruit"][1] = " Яблоко "
mixedSalad["fruit"][2] = " Банан "
mixedSalad["fruit"][3] = " Кокос "
mixedSalad["vegetables"][1] = "Салат-латук " 

mixedSalad["vegetables"][2] = "Помидор"
mixedSalad["vegetables"][3] = "Морковь"
TextWindow.WriteLine("-----")
For i = 1 To 3
  TextWindow.WriteLine("mixedSalad[fruit][" + i + "] содержит " + mixedSalad["fruit"][i])
EndFor
For i = 1 To 3
  TextWindow.WriteLine("mixedSalad[vegetables][" + i + "] содержит " + mixedSalad["vegetables"][i])
EndFor


Clock (Работа с ситемными часами)
Этот класс предоставляет доступ к системным часам.

Свойства
Date (This property is read-only.) 

Clock.Date - Устанавливает текущее системное время.
Day (This property is read-only.) 

Clock.Day - Получает текущий день месяца.
ElapsedMilliseconds (This property is read-only.)
Clock.ElapsedMilliseconds - Получает число миллисекунд, прошедших с 1900 года.

Hour (This property is read-only.) 

Clock.Hour - Получает текущий час.
Millisecond (This property is read-only.)

Clock.Millisecond - Возвращает текущую миллисекунду.

Minute (This property is read-only.) 

Clock.Minute - Получает текущую минуту.
Month (This property is read-only.) 

Clock.Month - Получает текущий месяц.
Second (This property is read-only.) 

Clock.Second - Получает текущую секунду.
Time (This property is read-only.) 

Clock.Time - Получает текущее системное время.
WeekDay (This property is read-only.) 

Clock.WeekDay - Получает текущий день недели.
Year (This property is read-only.) 

Clock.Year - Получает текущий год.
	‘Пример программы работы со стеком
TextWindow.WriteLine ("Сегодня (дата):" + Clock.Date + ".")
TextWindow.WriteLine ("Сегодня (месяц/день/год):" + Clock.Month + "/" + Clock.Day + "/" + Clock.Year + ".")
TextWindow.WriteLine ("Текущие время:" + Clock.Time + ".")
TextWindow.WriteLine ("Текущее время (часы:минуты:секунды):" + Clock.Hour + ":" + Clock.Minute + ":" + Clock.Second + ".")
TextWindow.WriteLine (Clock.ElapsedMilliseconds + "миллисекунд прошло с 1900 года.")
TextWindow.WriteLine ("Текущей миллисекундой является:" + Clock.Millisecond + ".")
TextWindow.WriteLine ("Текущий день недели:" + Clock.Weekday + ".")


Desktop (Работа с рабочим столом)
Этот класс обеспечивает взаимодействие с рабочим столом.
Свойства
Height (This property is read-only.) 

Desktop.Height - Возвращает значение высоты экрана на основном рабочем столе.
Width (This property is read-only.) 

Desktop.Width - Возвращает значение ширины экрана на основном рабочем столе.
Операции
SetWallPaper

Desktop.SetWallPaper(fileOrUrl) - Устанавливает указанное изображение в качестве обоев на рабочем столе. Это может быть локальный файл или файл, расположенный в сети или в Интернете.
fileOrUrl - Имя файла или URL-адрес картинки.
Возвращает – Ничего
	‘Работа с рабочим столом.
TextWindow.WriteLine("Размер рабочего стола " + Desktop.Width + "x" + Desktop.Height + ".")
Desktop.SetWallPaper("C:\Windows\Web\Wallpaper\Windows\img0.jpg")


Dictionary (Работа с интернет-словарем)
Этот класс предоставляет доступ к Интернет-словарю.
Операции
GetDefinition

Dictionary.GetDefinition(word) - Возвращает определение слова на английском языке.
word - Слово, определение которого запрашивается.
Возвращает - Определение или несколько определений слова.
GetDefinitionInFrench
Dictionary.GetDefinitionInFrench(word) - Возвращает определение слова на французском языке.
word - Слово, определение которого запрашивается.

Возвращает - Определение или несколько определений слова.
	' Работа с Интернет-словарем.
TextWindow.WriteLine(Dictionary.GetDefinition("banana"))
TextWindow.WriteLine(Dictionary.GetDefinitionInFrench("banana"))
Хочется обратить внимание что эти операции позволяют работать со словарями только на Английском и Французском языках.

ЭТО НЕ ПЕРЕВОДЧИК.


File (Работа с внешним файлом)
Объект File предоставляет методы для доступа к данным, чтения из файла и записи данных в файл. С помощью этого объекта настройки вашей программы могут быть сохранены перед завершением работы программы и загружены при следующем запуске.

Свойства
LastError

File.LastError - Получает или устанавливает сообщение о последней ошибке, случившейся при операциях с файлами. Это свойство полезно при выяснении причин сбоя какого-либо метода.
Операции
AppendContents

File.AppendContents(filePath, contents) - Открывает указанный файл и добавляет информацию в конец файла.
filePath - Полный путь к файлу, из которого нужно прочитать данные.  Например, полный путь может выглядеть так c:\temp\settings.data.
contents - Текст, который будет добавлен в конец файла.
Возвращает - В случае успешного выполнения операции возвращает "SUCCESS". В противном случае возвращает "FAILED".

CopyFile
File.CopyFile(sourceFilePath, destinationFilePath) - Копирует указанный исходный файл в заданное расположение. Если заданное расположение указывает на несуществующую папку, то метод попытается ее создать. Существующие файлы будут перезаписаны. Рекомендуется проверять, существуют ли целевые файлы, если вы не хотите их перезаписать.
sourceFilePath - Полный путь к файлу, который нужно скопировать. Например, полный путь может выглядеть так c:\temp\settings.data.
destinationFilePath - Путь, по которому надо скопировать файл.
Возвращает - В случае успешного выполнения операции возвращает "SUCCESS". В противном случае возвращает "FAILED".
CreateDirectory

File.CreateDirectory(directoryPath) - Создает указанную папку.
directoryPath - Полный путь к папке, которую надо создать.
Возвращает - В случае успешного выполнения операции возвращает "SUCCESS".  В противном случае возвращает "FAILED".
DeleteDirectory

File.DeleteDirectory(directoryPath) - Удаляет указанную папку.
directoryPath - Полный путь к каталогу, который должен быть удален.
Возвращает - В случае успешного выполнения операции возвращает "SUCCESS". В противном случае возвращает "FAILED".
DeleteFile
File.DeleteFile(filePath) - Удаляет указанный файл.
filePath - Путь к конечному месту расположения файла. Например, полный путь может выглядеть такc:\temp\settings.data.
Возвращает - В случае успешного выполнения операции возвращает "SUCCESS". В противном случае возвращает "FAILED".
GetDirectories

File.GetDirectories(directoryPath, arrayName) - Получает пути всех папок в указанной папке.
directoryPath - Папка для поиска вложенных папок.
arrayName - The name of the array that will be populated with the results. 
Возвращает - В случае успешного выполнения операция возвращает список папок в виде массива. В противном случае возвращает "FAILED".
GetFiles

File.GetFiles(directoryPath, arrayName) - Получает пути всех файлов в указанной папке.
directoryPath - Папка для поиска файлов.
arrayName - The name of the array that will be populated with the results. 
Возвращает - В случае успешного выполнения операция возвращает список файлов в виде массива. В противном случае возвращает "FAILED".
GetSettingsFilePath

File.GetSettingsFilePath() - Получает полный путь к файлу настроек для данной программы. Имя файла настроек программы формируется на основании имени программы и располагается в том же каталоге, что и сама программа.
Возвращает - Полный путь к файлу настроек для текущего приложения.
GetTemporaryFilePath

File.GetTemporaryFilePath() - Создает новый временный файл во временной папке и возвращает полный путь к файлу.
Returns - Полный путь ко временному файлу.
InsertLine
File.InsertLine(filePath, lineNumber, contents) - Открывает указанный файл и вставляет информацию по указанному номеру строки. Эта операция не перезаписывает существующую информацию в файле, а добавляет ее по указанному номеру строки.
filePath - Полный путь к файлу, из которого нужно прочитать данные.  Например, полный путь может выглядеть так c:\temp\settings.data.
lineNumber - Номер строки, по которому надо вставить текст.
Contents - Информация, которую надо вставить в файл.
Возвращает - В случае успешного выполнения операции возвращает "SUCCESS". В противном случае возвращает "FAILED".
ReadContents

File.ReadContents(filePath) - Открывает файл и считывает все его содержимое. Этот метод будет работать быстро для файлов размером меньше 1 МБ, но будет работать заметно медленнее на файлах больше 10 МБ.
filePath - Полный путь к файлу, который нужно прочитать.  Например: c:\temp\settings.data.
Возвращает – Содержимое файла.
ReadLine

File.ReadLine(filePath, lineNumber) - Открывает указанный файл и считывает информацию по указанному номеру строки.
filePath - Полный путь к файлу, из которого нужно прочитать данные.  Например, полный путь может выглядеть так c:\temp\settings.data.
lineNumber - Номер строки, с которой начнется чтение текста.
Возвращает - Текст из указанного файла, начиная с указанной строки.
WriteContents

File.WriteContents(filePath, contents) - Открывает файл и записывает в него указанную информацию с заменой исходной информации в файле.
filePath - Полный путь к файлу, в который нужно записать данные.  Например, полный путь может выглядеть так c:\temp\settings.data.
contents - Информация для записи в файл.
Возвращает - В случае успешного выполнения операции возвращает "SUCCESS". В противном случае возвращает "FAILED".
WriteLine

File.WriteLine(filePath, lineNumber, contents) - Открывает указанный файл и записывает информацию, находящуюся по указанному номеру строки. Эта операция перезаписывает информацию, находящуюся в файле по указанному номеру строки.
filePath - Полный путь к файлу, который нужно прочитать. Например, полный путь может выглядеть так c:\temp\settings.data.
lineNumber - Номер строки, с которой начнется запись текста.
Contents - Текст, который будет записан в файл, начиная с указанной строки.
Возвращает - В случае успешного выполнения операции возвращает "SUCCESS". В противном случае возвращает "FAILED".
	Как показано в следующем примере, вы можете использовать Microsoft Small Basic класс File для создания файлов, директорий, а также копировать файлы из одной директории в другую. Вы также можете прочитать информацию из файла и записать данные в файл.

' Этот код предполагает, что каталог C:\Temp\ уже существует.
TextWindow.WriteLine("Создание поддиректории: " + File.CreateDirectory("C:\Temp\TempSubdirectory\"))
subdirectories = File.GetDirectories("C:\Temp")
If Array.GetItemCount(subdirectories) = 1 Then
  TextWindow.WriteLine("C:\Temp имеет 1 поддиректорию.")
Else
  TextWindow.WriteLine("C:\Temp имеет " + Array.GetItemCount(subdirectories) + " поддиректорий.")
EndIf
For i = 1 To Array.GetItemCount(subdirectories)
  TextWindow.WriteLine("поддиректория[" + i + "] = " + subdirectories[i])
  files = File.GetFiles("C:\Temp\" + subdirectories[i])
  For j = 1 To Array.GetItemCount(files)
    TextWindow.WriteLine("    файлы[" + i + "] = " + files[i])
  EndFor
EndFor 

TextWindow.WriteLine("Путь к файлу этой программы " + File.GetSettingsFilePath())
filePath = File.GetTemporaryFilePath()
TextWindow.WriteLine("Временный файл был создан в: " + filePath + ".")
TextWindow.WriteLine("Записать в файл: " + File.WriteContents(filePath, "Привет, Мир!"))
TextWindow.WriteLine("Записать строку в файл: " + File.WriteLine(filePath, 15, "Снова привет, Мир!"))
TextWindow.WriteLine("Вставить строку в файл: " + File.InsertLine(filePath, 20, "Привет ещё раз, Мир!"))
TextWindow.WriteLine("Добавить в файл: " + File.AppendContents(filePath, "Привет в последний раз, Мир!"))
TextWindow.WriteLine("Файл содержит: " + File.ReadContents(filePath))
TextWindow.WriteLine("Копирует файл: " + File.CopyFile(filePath, "C:\Temp\TempSubdirectory\"))
If File.LastError = "" Then
Else
  TextWindow.WriteLine("Последняя ошибка была: " + File.LastError)
EndIf


Flickr (Доступ к Интернет-сервису фотографий)
Этот класс обеспечивает доступ к Интернет-сервису для размещения фотографий Flickr.
Операции
GetPictureOfMoment

Flickr.GetPictureOfMoment() - Получает URL-адрес для текущего изображения на сервисе Flickr.
Возвращает - URL-адрес файла текущего изображения на сервисе Flickr.
GetRandomPicture

Flickr.GetRandomPicture(tag) - Получает URL-адрес случайного изображение, которое помечено указанными тегами.
Tag - Тег для запрашиваемого изображения.
Возвращает - URL-адрес файла найденного изображения на сервисе Flickr.
	' Класс Flickr.
GraphicsWindow.Show()
' Обратите внимание, что операции Flickr имеют возможность выводить изображения, которые, возможно, не являются подходящими для младших пользователей.
GraphicsWindow.DrawImage(Flickr.GetPictureOfMoment(), 0, 0)
Program.Delay(5000)
GraphicsWindow.Clear()
GraphicsWindow.DrawImage( Flickr.GetRandomPicture("cat"), 0, 0)


GraphicsWindow (Работа в графическом окне)
Объект GraphicsWindow обеспечивает функционал ввода вывода графики. Например, с помощью этого класса можно рисовать и закрашивать окружности и прямоугольники.
Свойства
BackgroundColor

GraphicsWindow.BackgroundColor - Получает или устанавливает значение цвета фона окна.
BrushColor

GraphicsWindow.BrushColor - Читает или устанавливает цвет кисти (Brush) для заливки фигур в окне с графикой.
CanResize

GraphicsWindow.CanResize - Определяет, может ли пользователь изменять размер окна с графикой.
FontBold
GraphicsWindow.FontBold - Читает или устанавливает атрибут, который определяет является ли шрифт для вывода текста в окне с графикой жирным.
FontItalic
GraphicsWindow.FontItalic - Читает или устанавливает атрибут, который определяет является ли шрифт для вывода текста в окне с графикой курсивом.
FontName

GraphicsWindow.FontName - Получает или задает имя шрифта, который используется для вывода текста в окне.
FontSize

GraphicsWindow.FontSize - Получает или задает размер шрифта, который используется для вывода текста в окне.
Height

GraphicsWindow.Height - Читает или устанавливает высоту окна с графикой.
LastKey(This property is read-only.) 

GraphicsWindow.LastKey - Читает последнюю клавишу, которая была нажата или отпущена.
LastText(This property is read-only.) 

GraphicsWindow.LastText - Получает текст, который был введен последним в окно с графикой.
Left
GraphicsWindow.Left - Читает или устанавливает левую границу окна с графикой.
MouseX(This property is read-only.) 

GraphicsWindow.MouseX - Получает координату X для курсора мыши относительно окна с графикой.
MouseY(This property is read-only.) 

GraphicsWindow.MouseY - Получает координату Y для курсора мыши относительно окна с графикой.
PenColor
GraphicsWindow.PenColor - Получает или устанавливает значение цвета пера для рисования объектов в окне.
PenWidth
GraphicsWindow.PenWidth - Читает или устанавливает толщину пера (Pen), который рисует фигуры в окне с графикой.
Title
GraphicsWindow.Title - Читает или устанавливает заголовок окна с графикой.
Top
GraphicsWindow.Top - Читает или устанавливает верхнюю границу окна с графикой.
Width
GraphicsWindow.Width - Читает или устанавливает ширину окна с графикой.
События
KeyDown

GraphicsWindow.KeyDown - Происходит, когда пользователь нажимает клавишу на клавиатуре.
KeyUp

GraphicsWindow.KeyUp - Происходит, когда пользователь отпускает клавишу на клавиатуре.
MouseDown

GraphicsWindow.MouseDown - Вызывает событие при нажатии кнопки мыши.
MouseMove

GraphicsWindow.MouseMove - Вызывает событие при перемещении мыши.
MouseUp
GraphicsWindow.MouseUp - Вызывает событие при отпускании кнопки мыши.
TextInput

GraphicsWindow.TextInput - Вызывает событие при вводе текста в окно с графикой.
Операции
Clear
GraphicsWindow.Clear() - Очищает окно.
Возвращает - Ничего
DrawBoundText

GraphicsWindow.DrawBoundText(x, y, width, text) - Отображает текстовую строку на экране в заданном месте.
x - Координата X точки начала вывода текста.
y - Координата Y точки начала вывода текста.
width - Максимальная длина строки. Этот параметр определяет, где следует начать перенос текста.
text - Текст для отображения.
Возвращает - Ничего
DrawEllipse
GraphicsWindow.DrawEllipse(x, y, width, height) - Рисует на экране эллипс выбранным пером (Pen).
x - Координата X верхнего левого угла прямоугольника, в который вписывается эллипс.
y - Координата Y верхнего левого угла прямоугольника, в который вписывается эллипс.
Width - Значение ширины эллипса.
Height - Значение высоты эллипса.
Возвращает - Ничего
DrawImage

GraphicsWindow.DrawImage(imageName, x, y) - Отрисовывает на экране хранящееся в памяти изображение.
imageName - Имя изображения для отображения.
x - Координата X для отрисовки изображения.
y - Координата Y для отрисовки изображения.
Возвращает - Ничего
DrawLine

GraphicsWindow.DrawLine(x1, y1, x2, y2) - Рисует линию, соединяющую две точки.
x1 - Координата X первой точки.
y1 - Координата Y первой точки.
x2 - Координата X второй точки.
y2 - Координата Y второй точки.
Возвращает - Ничего
DrawRectangle

GraphicsWindow.DrawRectangle(x, y, width, height) - Рисует на экране прямоугольник выбранным пером (Pen).
x - Координата X верхнего левого угла прямоугольника.
y - Координата Y верхнего левого угла прямоугольника.
Width - Ширина прямоугольника.
Height - Высота прямоугольника.
Возвращает - Ничего
DrawResizedImage

GraphicsWindow.DrawResizedImage(imageName, x, y, width, height) - Отображает на экране хранящийся в памяти рисунок и меняет его размер в соответствии с заданным.
imageName - Имя изображения для отрисовки.
x - Координата X для отрисовки изображения.
y - Координата Y для отрисовки изображения.
Width - Ширина изображения для отрисовки.
Height - Ширина изображения для отрисовки.
Возвращает - Ничего
DrawText

GraphicsWindow.DrawText(x, y, text) - Отображает текстовую строку на экране в заданном месте.
x - Координата X точки начала вывода текста.
y - Координата Y точки начала вывода текста.
Text - Текст для вывода на экран.
Возвращает - Ничего
DrawTriangle

GraphicsWindow.DrawTriangle(x1, y1, x2, y2, x3, y3) - Рисует на экране треугольник выбранным пером (Pen).
x1 - Координата X первой вершины треугольника.
y1 - Координата Y первой вершины треугольника.
x2 - Координата X второй вершины треугольника.
y2 - Координата Y второй вершины треугольника.
x3 - Координата X третьей вершины треугольника.
y3 - Координата Y третьей вершины треугольника.
Возвращает - Ничего
FillEllipse

GraphicsWindow.FillEllipse(x, y, width, height) - Заполняет внутреннюю часть эллипса на экране с помощью выбранной кисти (Brush).
x - Координата X верхнего левого угла прямоугольника, в который вписывается эллипс.
y - Координата Y верхнего левого угла прямоугольника, в который вписывается эллипс.
Width - Значение ширины эллипса.
Height - Значение высоты эллипса.
Возвращает - Ничего
FillRectangle

GraphicsWindow.FillRectangle(x, y, width, height) - Заполняет внутреннюю часть прямоугольника на экране с помощью выбранной кисти (Brush).
x - Координата X верхнего левого угла прямоугольника.
y - Координата Y верхнего левого угла прямоугольника.
Width - Ширина прямоугольника.
Height - Высота прямоугольника.

Возвращает - Ничего
FillTriangle

GraphicsWindow.FillTriangle(x1, y1, x2, y2, x3, y3) - Рисует на экране треугольник и заполняет его внутреннюю часть с помощью выбранной кисти (Brush).
x1 - Координата X первой вершины треугольника.
y1 - Координата Y первой вершины треугольника.
x2 - Координата X второй вершины треугольника.
y2 - Координата Y второй вершины треугольника.
x3 - Координата X третьей вершины треугольника.
y3 - Координата Y третьей вершины треугольника.
Возвращает - Ничего
GetColorFromRGB
GraphicsWindow.GetColorFromRGB(red, green, blue) - Создает цвет, основываясь на заданных значениях красного, зеленого и синего цветов (RGB).
red - Значение красного компонента в цвете (0-255).
green - Значение зеленого компонента в цвете (0-255).
blue - Значение синего компонента в цвете (0-255)
Возвращает - Возвращает цвет, который может использоваться для задания цвета кисти или пера.
GetPixel
GraphicsWindow.GetPixel(x, y) - Возвращает цвет пикселя в точке с координатами X и Y.
x - Координата X пикселя.
y - Координата Y пикселя.
Возвращает - Цвет пикселя.
GetRandomColor

GraphicsWindow.GetRandomColor() - Получает цвет случайным образом.
Возвращает - Цвет, выбранный случайным образом.
Hide
GraphicsWindow.Hide() - Делает окно с графикой невидимым.
Возвращает - Ничего
SetPixel
GraphicsWindow.SetPixel(x, y, color) - Рисует пиксель заданного цвета в точке с координатами X и Y.
x - Координата X пикселя.
y - Координата Y пикселя.
Color  - Цвет пикселя
Возвращает - Ничего
Show
GraphicsWindow.Show() - Показывает окно с графикой.
Возвращает - Ничего
ShowMessage
GraphicsWindow.ShowMessage(text, title) - Отображает окно сообщения для пользователя.
text - Текст для отображения в окне сообщения.
title - Заголовок окна сообщения.
Возвращает – Ничего
	Microsoft Small Basic в графическом окне позволяет Вам делать следующие операции, показанные в примере кода:

· Изменять цвет фона окна, высоту, ширину, размещение по отношению к верхнему левому углу экрана, название и может ли пользователь изменять размер окна графики.

· Рисовать полые и сплошные эллипсы, прямоугольники и треугольники разного цвета.

· Просматривать изображения.

· Изменять размеры существующих фигур и изображений.

· Получать и изменять цвет отдельных пикселей.

· Реагировать на действия пользователя, такие как нажатие и отпускание клавиши на клавиатуре, перемещение мыши и нажатия и отпусканя кнопки мыши. 

' Создайте окно графики 700x500 смещенное от верхнего левого угла экрана 50, 100.
' Измените заголовок графического и сделайте цвет его фона зеленым.
' Позвольте пользователям изменять размеры окна графики. 
GraphicsWindow.BackgroundColor = "Green"
GraphicsWindow.CanResize = "True"
GraphicsWindow.Height = 500
GraphicsWindow.Width = 700
GraphicsWindow.Top = 50
GraphicsWindow.Left = 100
GraphicsWindow.Title = "Please Wait..."
GraphicsWindow.Clear()
' Выведите случайное изображение в графическом окне.
' Обратите внимание, что операции Flickr имеют возможность выводить изображения, которые, возможно, не являются подходящими для младших пользователей.
GraphicsWindow.DrawImage(Flickr.GetRandomPicture("cat"), 175, 0)
' Нарисуйте эллипс, линию, прямоугольник, текст, изображение и треугольник в графическом окне.
' Используйте синее перо. Не закрашивайте фигуры.
GraphicsWindow.PenColor = "Blue"
GraphicsWindow.PenWidth = 5
GraphicsWindow.DrawEllipse(100, 150, 50, 20)
GraphicsWindow.DrawLine(0, 0, 50, 100)
GraphicsWindow.DrawRectangle(25, 175, 60, 30)
GraphicsWindow.DrawBoundText(0, 20, GraphicsWindow.Width, "Hello, World!")
' Обратите внимание, что операции Flickr имеют возможность выводить изображения, которые, возможно, не являются подходящими для младших пользователей.
GraphicsWindow.DrawResizedImage(Flickr.GetRandomPicture("dog"), 10, 250, 100, 75)
GraphicsWindow.DrawTriangle(15, 350, 55, 350, 35, 375)
' Нарисуйте закрашенный эллипс, прямоугольник и треугольник.
' Используйте специальную цветную палитру "красный зеленый синий цвет" (RGB).
GraphicsWindow.BrushColor = GraphicsWindow.GetColorFromRGB(125, 100, 75)
GraphicsWindow.FillEllipse(10, 400, 50, 20)
' Используйте цвет пикселя в местоположении окна графики 0, 10.
GraphicsWindow.BrushColor = GraphicsWindow.GetPixel(0, 10)
GraphicsWindow.FillRectangle(5, 455, 60, 30)
GraphicsWindow.FillTriangle(100, 455, 150, 425, 125, 475)
' Измените цвет фона графического окна на случайный цвет.
GraphicsWindow.BackgroundColor = GraphicsWindow.GetRandomColor()
' Измените цветовую палитру фона индивидуальных пикселей к случайному цвету.
For x = 70 to 170
  For y = 40 to 80
    GraphicsWindow.SetPixel(x, y, GraphicsWindow.GetRandomColor())
  EndFor
EndFor
GraphicsWindow.Title = "Моё графическое окно"
' Покажите диалоговое окно.
GraphicsWindow.ShowMessage("Смотрите, случайные цвета! ", "Набор пикселей")
' Позвольте графическому окну реагировать на действия мыши и клавиатуры.
GraphicsWindow.MouseDown = OnMouseDown
GraphicsWindow.MouseUp = OnMouseUp
GraphicsWindow.MouseMove = OnMouseMove
GraphicsWindow.KeyDown = OnKeyDown
GraphicsWindow.KeyUp = OnKeyUp
GraphicsWindow.TextInput = OnTextInput
Sub OnKeyDown
  GraphicsWindow.Title = "Эта " + GraphicsWindow.LastKey + " клавиша была нажата."
EndSub
Sub OnKeyUp
  GraphicsWindow.Title = "Эта " + GraphicsWindow.LastKey + " клавиша была отпущена."
EndSub
Sub OnMouseDown
  GraphicsWindow.Title = "Кнопка мыши была нажата."
EndSub
Sub OnMouseUp
  GraphicsWindow.Title = "Кнопка мыши была отпущена."
EndSub
Sub OnMouseMove
  GraphicsWindow.Title = "Мышь переместилась в " + GraphicsWindow.MouseX + ", " + GraphicsWindow.MouseY + "."
EndSub
Sub OnTextInput
  GraphicsWindow.Title = "Любой введенный текст."
EndSub


ImageList (Работа с внешним изображением)
Этот класс помогает загружать и хранить изображения в памяти.
Операции
GetHeightOfImage
ImageList.GetHeightOfImage(imageName) - Получает высоту хранимого изображения.
imageName - Имя изображения в памяти.
Возвращает - Высота указанного изображения.
GetWidthOfImage
ImageList.GetWidthOfImage(imageName) - Получает ширину хранимого изображения.
imageName - Имя изображения в памяти.
Возвращает - Ширина указанного изображения.
LoadImage
ImageList.LoadImage(fileNameOrUrl) - Загружает изображение из файла или из Интернета в память.
fileNameOrUrl - Имя файла, из которого надо загрузить изображение.  Это может быть как локальный файл, так и адрес в сети Интернет.
Возвращает – Ничего
	' Работа с внешним изображением.
URL = Flickr.GetRandomPicture("cat")
' Обратите внимание, что операции LoadImage имеют возможность выводить изображения, которые, возможно, не являются подходящими для младших пользователей.
GraphicsWindow.DrawImage(ImageList.LoadImage(URL), 0, 0)
GraphicsWindow.Width = ImageList.GetWidthOfImage(URL)
GraphicsWindow.Height = ImageList.GetHeightOfImage(URL)


Math (Работа с математическими операторами)
Класс Math предоставляет множество полезных методов для математических операций.
Свойства
Pi (This property is read-only.) 

Math.Pi - Получает значение Пи.
Операции
Abs
Math.Abs(number) - Возвращает  абсолютное  значение  для  заданного  числа.   Например,   для (-32.233) метод вернет (32.233).

number - Число, абсолютное значение которого надо получить.
Возвращает - Абсолютное значение заданного числа.
ArcCos
Math.ArcCos(cosValue) - Возвращает значение угла в радианах, соответствующее указанному значению косинуса.

cosValue - Значение косинуса, для которого надо получить угол.
Возвращает - Угол (в радианах) для заданного значения косинуса.
ArcSin

Math.ArcSin(sinValue) - Возвращает значение угла в радианах, соответствующее указанному значению синуса.

sinValue - Значение синуса, для которого надо получить угол.

Возвращает - Угол (в радианах) для заданного значения синуса.

ArcTan

Math. ArcTan(tanValue) Получает угол в радианах для заданного значения тангенса.

tanValue - Значения тангенса, для которого надо получить угол.

Возвращает - Угол (в радианах) для заданного значения тангенса.

Ceiling
Math.Ceiling(number) - Округляет заданное десятичное значение до большего или равного целого числа. Например, для 32,233 метод вернет 33.

number - Число, которое надо округлить в большую сторону.
Возвращает - Число, округленное в большую сторону.
Cos
Math.Cos(angle) - Получает значение косинуса для заданного в радианах угла.

angle - Значение угла (в радианах), чей косинус требуется вычислить.
Возвращает - Значение косинуса данного угла.
Floor
Math.Floor(number) - Округляет заданное десятичное значение до меньшего или равного целого числа.  Например, для 32,233 метод вернет 32.

number - Число, которое надо округлить в меньшую сторону.
Возвращает - Число, округленное в меньшую сторону.
GetDegrees
Math.GetDegrees(angle) - Конвертирует заданный угол из радиан в градусы.

angle - Значение угла в радианах.
Возвращает - Значение угла, переведенное в градусы.
GetRadians
Math.GetRadians(angle) - Конвертирует заданный угол из градусов в радианы.

angle - Значение угла в градусах.
Возвращает - Значение угла, переведенное в радианы.
GetRandomNumber
Math.GetRandomNumber(maxNumber) - Возвращает случайное число в диапазоне между 1 и числом, заданным в параметре maxNumber (включительно).

maxNumber - Максимальное значение для диапазона случайных чисел.
Возвращает - Случайное число, которое меньше или равно указанному максимальному значению.
Log
Math.Log(number) - Получает десятичный логарифм для заданного числа.

number - Число для вычисления десятичного логарифма.
Возвращает - Десятичный логарифм для указанного числа.
Max
Math.Max(number1, number2) - Сравнивает два значение и возвращает большее из них.
number1 - Первое из сравниваемых чисел.
number2 - Второе из сравниваемых чисел.
Возвращает - Наибольшее значение из двух заданных чисел.
Min
Math.Min(number1, number2) - Сравнивает два значение и возвращает меньшее из них.
number1 - Первое из сравниваемых чисел.
number2 - Второе из сравниваемых чисел.

Возвращает - Наименьшее значение из двух заданных чисел.
NaturalLog
Math.NaturalLog(number) - Получает натуральный логарифм для заданного числа.

number - Число, для которого надо вычислить натуральный логарифм.
Возвращает - Значение натурального логарифма для заданного числа.
Power
Math.Power(baseNumber, exponent) - Возводит число в указанную степень.
baseNumber - Число для возведения в степень.
exponent - Степень, в которую надо возвести число.
Возвращает - Число, возведенное в указанную степень.
Remainder
Math.Remainder(dividend, divisor) - Делит первое число на второе и возвращает остаток.
dividend - Делимое.
divisor - Делитель.
Возвращает - Остаток от деления.
Round
Math.Round(number) - Округляет заданное десятичное значение до ближайшего целого числа.  Например, 32.233 будет округлено до 32.0, а 32.566 до 33.

number - Число для округления.
Возвращает - Округленное значение заданного числа.
Sin
Math.Sin(angle) - Получает значение синуса для заданного в радианах угла.

angle - Угол для вычисления синуса (в радианах).
Возвращает - Синус для заданного угла.
SquareRoot
Math.SquareRoot(number) - Получает квадратный корень для заданного числа.

number - Число, квадратный корень которого надо вычислить.
Возвращает - Квадратный корень для заданного числа.
Tan
Math.Tan(angle) - Получает значение тангенса для заданного в радианах угла.

angle - Значение угла (в радианах), чей тангенс требуется вычислить.
Возвращает - Значение тангенса данного угла.
	Microsoft Small Basic поддерживает несколько математических функций, включая не только простое суммирование, вычитание, умножение и деление, но также и сравнение, случайные числа, округление, вычисление экспоненты, логарифма, Пи, дуги и тригонометрические функции, как демонстрируется в следующем коде:
TextWindow.WriteLine("Большее из чисел 10 и 15 это: " + Math.Max(15, 10))
TextWindow.WriteLine("Меньшее из чиселномер 10 и 15 это: " + Math.Min(15, 10))
TextWindow.WriteLine("Случайное число между 1 и 10 это: " + Math.GetRandomNumber(10))
TextWindow.WriteLine("Абсолютное значение -500 это: " + Math.Abs(-500))
TextWindow.WriteLine("Округление в большую сторону 4.02 это: " + Math.Ceiling(4.02))
TextWindow.WriteLine("Округление в меньшую сторону 4.02 это: " + Math.Floor(4.02))
TextWindow.WriteLine("Значение Пи это: " + Math.Pi)
TextWindow.WriteLine("5 возведенное в 3 степень это: " + Math.Power(5, 3))
TextWindow.WriteLine("Квадратный корень из 25 это: " + Math.SquareRoot(25))
TextWindow.WriteLine("Возвращенный остаток при делении 10 на 4 это: " + Math.Remainder(10, 4))
TextWindow.WriteLine("Результат округления 4.02 до ближайшего целого это: " + Math.Round(4.02))
TextWindow.WriteLine("Десятичный логарифм от 1000 это: " + Math.Log(1000))
TextWindow.WriteLine("Натуральный логарифм от 1000 это: " + Math.NaturalLog(1000))
TextWindow.WriteLine("-----")
' Радиан определяется как угловая величина дуги, длина которой равна её радиусу.
' Угол 180 градусов заканчивается длиной дуги, равной Пи.
TextWindow.WriteLine("1 радиан = " + Math.GetDegrees(1) + " градусов (180/Pi).")
TextWindow.WriteLine("90 градусов = " + Math.GetRadians(90) + " радиан (Pi/2).")
TextWindow.WriteLine("180 градусов = " + Math.GetRadians(180) + " радиан (Pi).")
TextWindow.WriteLine("Давайте проверим нашу работу...")
TextWindow.WriteLine("180/Pi = " + 180/Math.Pi + " градусов.")
TextWindow.WriteLine("Pi/2 = " + Math.Pi/2 + " радиан.")
TextWindow.WriteLine("Pi = " + Math.Pi + " радиан.")
TextWindow.WriteLine("-----")
' Рассматривая прямоугольный треугольник с углами равными 45 градусам и сторонами имеющими длины 1, 1 и ~1.414: 
TextWindow.WriteLine("Синус угла 45 градусов это: " + Math.Sin(Math.GetRadians(45)) + ", или 1/~1.414.")
TextWindow.WriteLine("Косинус угла 45 градусов это: " + Math.Cos(Math.GetRadians(45)) + ", или 1/~1.414.")
TextWindow.WriteLine("Тангес угла 45 градусов это: " + Math.Tan(Math.GetRadians(45)) + ", или 1/1.")
TextWindow.WriteLine("Давайте проверим нашу работу...")
TextWindow.WriteLine("1/" + Math.SquareRoot(2) + " = " + 1/Math.SquareRoot(2))
TextWindow.WriteLine("-----")
' Рассматривая прямоугольный треугольник с углами равными 63,5 и 26,5 градусам и сторонами имеющими длины 1, 2 и ~2.24:
TextWindow.WriteLine("Синус угла 63.5 градуса это: " + Math.Sin(Math.GetRadians(63.5)) + ", или 2/~2.24.")
TextWindow.WriteLine("Косинус угла 63.5 градусов это: " + Math.Cos(Math.GetRadians(63.5)) + ", или 1/~2.24.")
TextWindow.WriteLine("Тангес угла 63.5 градусов это: " + Math.Tan(Math.GetRadians(63.5)) + ", или 2/1.")
TextWindow.WriteLine("Давайте проверим нашу работу...")
TextWindow.WriteLine("2/" + Math.SquareRoot(5) + " = " + 2/Math.SquareRoot(5))
TextWindow.WriteLine("1/" + Math.SquareRoot(5) + " = " + 1/Math.SquareRoot(5))
TextWindow.WriteLine("-----")


Mouse (Работа со свойствами мыши)
Класс Mouse предоставляет методы для получения или установки свойств мыши, таких как положение курсора, тип курсора и т.д.
Свойства
IsLeftButtonDown (This property is read-only.) 

Mouse.IsLeftButtonDown - Определяет была ли нажата левая кнопка мыши.
IsRightButtonDown (This property is read-only.) 

Mouse.IsRightButtonDown - Определяет была ли нажата правая кнопка мыши.
MouseX 

Mouse.MouseX - Получает или устанавливает координату X курсора мыши.
MouseY 

Mouse.MouseY - Получает или устанавливает координату Y курсора мыши.
Операции
HideCursor
Mouse.HideCursor() - Скрывает курсор мыши на экране.
Возвращает - Ничего
ShowCursor
Mouse.ShowCursor() - Показывает курсор мыши на экране.
Возвращает – Ничего
	Если Вы заинтересованы созданием в основном текстовых программ, Microsoft Small Basic имеет классы TextWindow и Text, вероятно, достаточно для ваших потребностей. Однако, если вы заинтересованы в интерактивных графических программах (например, играми или личными приложениями повышающих Вашу производительность), Microsoft Small Basic имеет класс GraphicsWindow, который поддерживает работу мыши и клавиатуры. Чтобы узнать, как это работает, можно попробовать импортировать следующий код в Small Basic:

' Стандартный код для того, чтобы соединить действия с клавиатурой, мышью и вводимом текстом действия в графическом окне.
GraphicsWindow.KeyDown = OnKeyDown
GraphicsWindow.KeyUp = OnKeyUp
GraphicsWindow.MouseDown = OnMouseDown
GraphicsWindow.MouseMove = OnMouseMove
GraphicsWindow.MouseUp = OnMouseUp
GraphicsWindow.TextInput = OnTextInput
Sub OnKeyDown
  ' Код при нажатии клавиши расположен здесь.
  GraphicsWindow.Title = "'" + GraphicsWindow.LastKey + "' нажата" 

EndSub
Sub OnKeyUp
  ' Код при отпускании клавиши расположен здесь.
  GraphicsWindow.Title = "'" + GraphicsWindow.LastKey + "' отпущена"
EndSub
Sub OnMouseDown
  ' Код при нажатии клавиши мыши расположен здесь.
  If Mouse.IsLeftButtonDown Then
    GraphicsWindow.Title = "Левая кнопка нажата"
  ElseIf Mouse.IsRightButtonDown Then
    GraphicsWindow.Title = "Правая кнопка нажата"
  Else
    GraphicsWindow.Title = "Нажата другая кнопка мыши (кроме левой и правой)"
  EndIf
EndSub
Sub OnMouseMove
  ' Код придвижении мышью расположен здесь.
  ' GraphicsWindow.MouseX и GraphicsWindow.MouseY координаты курсора мыши относительно графического окна.
  ' Mouse.MouseX и Mouse.MouseY координаты курсора мыши относительно всего экрана (Рабочего стола), который обычно больше чем окно графики.
  GraphicsWindow.Title = "GWX = " + GraphicsWindow.MouseX + ", GWY = " + GraphicsWindow.MouseY + ", ЭкранныйX = " + Mouse.MouseX + ", ЭкранныйY = " + Mouse.MouseY
EndSub
Sub OnMouseUp
  ' Код при отпускании кнопки мыши расположен здесь.
  GraphicsWindow.Title = "Любая кнопка мыши отпущена"
EndSub
Sub OnTextInput
  ' Код для текстового ввода расположен здесь.
EndSub
Обратите внимание, что при перемещении мыши, нажатии и отпускании кнопки мыши, нажатии и отпускании клавиши на клавиатуре графическое окно регистрирует информацию о мыши и ключевых событий. Вы можете использовать эту информацию, чтобы Ваша программа принимала определенные действия в ответ на эти события.


Network (Работа с сетью)
Этот класс предоставляет доступ к сетевым методам.

Операции
DownloadFile
Network.DownloadFile(url) - Загружает файл из сети в локальный временный файл.

url - Путь к файлу в сети.
Возвращает - Локальное имя файла, который был загружен из сети.
GetWebPageContents
Network.GetWebPageContents(url) - Получает содержимое указанной веб-страницы.

url - URL-адрес веб-страницы
Возвращает - Содержание указанной веб-страницы.
	' Работа с сетью.
' Обратите внимание, что операции Network имеют возможность выводить контент, который, возможно, не являются подходящими для младших пользователей. TextWindow.WriteLine("Downloaded a file to: " + Network.DownloadFile("http://msdn.microsoft.com/en-us/default.aspx"))
GraphicsWindow.DrawBoundText(0, 0, GraphicsWindow.Width, Network.GetWebPageContents("http://msdn.microsoft.com/en-us/default.aspx"))


Program (Контроль над выполнением программы)
Класс Program предоставляет вспомогательные методы для контроля выполнения программы.
Свойства
ArgumentCount (This property is read-only.)
Program.ArgumentCount - Получает количество аргументов командной строки, переданных программе.
Directory (This property is read-only.) 

Program.Directory - Получает папку для запущенной программы.
Операции
Delay
Program.Delay(milliSeconds) - Задерживает выполнение программы на указанное число миллисекунд.

milliSeconds - Время задержки.
Возвращает - Ничего
End
Program.End() - Заканчивает программу.
Возвращает - Ничего
GetArgument

Program.GetArgument(index) - Возвращает указанный аргумент из тех, что были переданы в текущую программу.


Index - Индекс (порядковый номер) аргумента.

Возвращает - Аргумент командной строки, расположенный по указанному индексу.
	' Контроль над выполнением программы.
' Если Вы запускаете программу Small Basic из командной строки, Вы можете передать ей аргументы.
' (Ищите *.exe файл в том же самом каталоге где сохранен соответствующий *.sb файл).
TextWindow.WriteLine("Количество аргументов это: " + Program.ArgumentCount + ".")
TextWindow.WriteLine("Директория программы: " + Program.Directory + ".")
TextWindow.WriteLine("Первый аргумент это: " + Program.GetArgument(1) + ".")
Program.Delay(2000)
Program.End()


Shapes (Работа с фигурами)
Объект Shape (фигура) позволяет добавлять, двигать и вращать фигуры в графическом окне.

Операции

AddEllipse
Shapes.AddEllipse(width, height) - Добавляет эллипс с заданной шириной и высотой.
Width - Ширина эллипса.
Height - Высота эллипса.
Возвращает - Эллипс, который только что был добавлен в Графическое Окно.
AddImage

Shapes.AddImage(imageName) -Добавляет изображение в качестве фигуры, которую можно перемещать, анимировать или вращать.
imageName - Имя фигуры для отображения.
Возвращает - Изображение, которое только что было добавлено в графическому окно.

AddLine
Shapes.AddLine(x1, y1, x2, y2) - Добавляет линию, соединяющую две заданные точки.
x1 - Координата X первой точки.
y1 - Координата Y первой точки.
x2 - Координата X второй точки.

y2 - Координата Y второй точки.
Возвращает - Линия, которая была только что добавлена в Графическое Окно.
AddRectangle

Shapes.AddRectangle(width, height) - Добавляет прямоугольник с заданной шириной и выстой.
Width - Ширина прямоугольника.
Height - Высота прямоугольника.

Возвращает - Прямоугольник, который был только что добавлен в Графическое окно.
AddTriangle

Shapes.AddTriangle(x1, y1, x2, y2, x3, y3) - Добавляет треугольник, который представлен в виде набора вершин.
x1 - Координата X первой вершины треугольника.
y1 - Координата Y первой вершины треугольника.
x2 - Координата X второй вершины треугольника.
y2 - Координата Y второй вершины треугольника.
x3 - Координата X третьей вершины треугольника.
y3 - Координата Y третьей вершины треугольника.
Возвращает - Треугольник, который был только что добавлен в Графическое Окно.
Animate
Shapes.AnimateShape(shapeName, x, y, duration) - Перемещает фигуру с заданным именем на новую позицию с использованием анимации.
shapeName - Имя перемещаемой фигуры.
x - Координата X новой позиции.
y - Координата Y новой позиции.
Duration - Время анимации в миллисекундах.
Возвращает - Ничего
GetLeft
Shapes.GetLeft(shapeName) - Получает координаты самой левой точки фигуры.

shapeName - Имя фигуры.
Возвращает - Координаты самой левой точки фигуры.
GetOpacity

Shapes.GetOpacity(shapeName) - Получает уровень прозрачности фигуры.  


shapeName - Имя фигуры.

Возвращает - Уровень прозрачности фигуры в виде значения от 0 до 100, где 0 - полностью прозрачная и 100 - полностью непрозрачная.

GetTop
Shapes.GetTop(shapeName) - Получает координаты самой верхней точки фигуры.

shapeName - Имя фигуры.
Возвращает - Координаты самой верхней точки фигуры.
HideShape

Shapes.HideShape(shapeName) - Скрывает уже добавленную фигуру.


shapeName - Имя фигуры.

Возвращает - Ничего

Move
Shapes.Move(shapeName, x, y) - Перемещает фигуру с заданным именем в новою позицию.
shapeName - Имя фигуры, которую нужно переместить.
x - Координата X новой позиции.
y - Координата Y новой позиции.
Возвращает - Ничего

Remove
Shapes.Remove(shapeName) - Удаляет фигуру из окна с графикой.

shapeName - Имя фигуры, которую нужно удалить.
Возвращает - Ничего
Rotate
Shapes.Rotate(shapeName, angle) - Поворачивает фигуру с заданным именем на заданное число градусов.
shapeName - Имя поворачиваемой фигуры.
angle - Угол, на который надо повернуть фигуру.
Возвращает – Ничего
SetOpacity

Shapes.SetOpacity(shapeName, level) - Устанавливает, как будет отображаться уровень прозрачности.

shapeName - Имя фигуры.

level - Уровень прозрачности фигуры в виде значения от 0 до 100, где 0 - полностью прозрачная и 100 - полностью непрозрачная.

Возвращает - Ничего

ShowShape

Shapes.ShowShape(shapeName) - Отображает скрытую фигуру.

shapeName - Имя фигуры.

Возвращает – Ничего

Zoom

Shapes.Zoom(shapeName, scaleX, scaleY) - Масштабирует фигуру, используя заданный уровень увеличения (от 0,1 до 20).

shapeName - Имя поворачиваемой фигуры.

scaleX - Уровень масштабирования по оси X.

scaleY - Уровень масштабирования по оси Y.
Возвращает – Ничего
	Microsoft Small Basic класс фигуры позволяет сделать следующие операции с фигурами в графическом окне, которые показаны в следующем коде: 

· Добавлять фигуры, такие как эллипсы, изображения, линии, прямоугольники и треугольники. Эти фигуры можно анимировать, перемещать, вращать и масштабировать.

· Получать экранные координаты существующих фигур.

· Скрывать и снова отображать существующие фигуры.

· Получать и изменять непрозрачность ("прозрачность") существующей фигуры.

ellipse = Shapes.AddEllipse(50, 30)
image = Shapes.AddImage(Flickr.GetRandomPicture("cat"))
' Приостанавливаем программу после каждого действия, чтобы Вы могли видеть то, что получается.
PauseProgram()
Shapes.Move(image, 200, 300)
PauseProgram()
line = Shapes.AddLine(0, 100, 20, 120)
PauseProgram()
rectangle = Shapes.AddRectangle(50, 30)
PauseProgram()
triangle = Shapes.AddTriangle(0, 0, 0, 30, 30, 0)
PauseProgram()
Shapes.Animate(ellipse, 100, 100, 2000)
PauseProgram()
GraphicsWindow.Title = "лево линии это: " + Shapes.GetLeft(line)
PauseProgram()
GraphicsWindow.Title = "верх эллипса это: " + Shapes.GetTop(ellipse)
PauseProgram()
Shapes.HideShape(triangle)
PauseProgram()
Shapes.Move(rectangle, 100, 10)
PauseProgram()
Shapes.Remove(line)
PauseProgram()
Shapes.Rotate(rectangle, 45)
PauseProgram()
Shapes.SetOpacity(rectangle, 50)
PauseProgram()
Shapes.ShowShape(triangle)
PauseProgram()
Shapes.Zoom(ellipse, 3, 5)
GraphicsWindow.ShowMessage("Готово!", "Готово")
Sub PauseProgram
  ' Приостановка программы на 2 секунды (2000 миллисекунд).
  Program.Delay(2000)


Sound (Работа со звуком)
Объект Sound предоставляет методы для воспроизведения аудиофайлов. К библиотеке прилагаются образцы аудиофайлы.

Операции
Pause
Sound.Pause(filePath) - Приостанавливает воспроизведение (режим паузы) аудиофайла. Если файл в этот момент не проигрывался, операция ничего не делает.
filePath - Путь к аудиофайлу. Это может быть как локальный файл (например, c:\music\track1.mp3), так и файл, находящийся в сети (например, http://contoso.com/track01.wma).
Возвращает - Ничего
Play
Sound.Play(filePath) - Проигрывает аудиофайл. Это может быть файл формата mp3, wav или wma. Другие форматы файлов могут требовать наличия соответствующих аудиокодеков на компьютере пользователя. Если файл был уже поставлен на паузу, то эта операция продолжит воспроизведение с места его приостановки.
filePath - Путь к аудиофайлу. Это может быть как локальный файл (например, c:\music\track1.mp3), так и файл, находящийся в сети (например, http://contoso.com/track01.wma).

Возвращает - Ничего
PlayAndWait
Sound.PlayAndWait(filePath) - Проигрывает аудиофайл и ожидает окончания воспроизведения.  Это может быть файл формата mp3, wav или wma. Другие форматы файлов могут требовать наличия соответствующих аудиокодеков на компьютере пользователя. Если файл был поставлен на паузу, то эта операция продолжит воспроизведение с места его приостановки.
filePath - Путь к аудиофайлу. Это может быть как локальный файл (например, c:\music\track1.mp3), так и файл, находящийся в сети (например, http://contoso.com/track01.wma).

Возвращает - Ничего
PlayBellRing
Sound.PlayBellRing() - Проигрывает звук звонка.
Возвращает - Ничего
PlayBellRingAndWait
Sound.PlayBellRingAndWait() - Проигрывает звук звонка и ждет, пока он не закончится.
Возвращает - Ничего
PlayChime
Sound.PlayChime() - Проигрывает звук колокола.
Возвращает - Ничего
PlayChimeAndWait
Sound.PlayChimeAndWait() - Проигрывает звук колокола и ждет, пока он не закончится.
Возвращает - Ничего
PlayChimes
Sound.PlayChimes() - Проигрывает звук колоколов.
Возвращает - Ничего
PlayChimesAndWait
Sound.PlayChimesAndWait() - Проигрывает звук колоколов и ждет, пока он не закончится.
Возвращает - Ничего
PlayClick
Sound.PlayClick() - Проигрывает звук щелчка мыши.
Возвращает - Ничего
PlayClickAndWait
Sound.PlayClickAndWait() - Проигрывает звук щелчка мыши и ждет, пока он не закончится.
Возвращает - Ничего
Stop
Sound.Stop(filePath) - Останавливает воспроизведение аудиофайла. Если файл в этот момент не проигрывался, операция ничего не делает.
filePath - Путь к аудиофайлу. Это может быть как локальный файл (например, c:\music\track1.mp3), так и файл, находящийся в сети (например, http://contoso.com/track01.wma).

Возвращает – Ничего
	' Работа со звуком.
WINDOWS_MEDIA_PATH = "C:\Windows\Media\"
Sound.Play(WINDOWS_MEDIA_PATH + "Windows Notify.wav")
DelayProgram()
Sound.PlayAndWait(WINDOWS_MEDIA_PATH + "Windows Exclamation.wav")
DelayProgram()
Sound.PlayBellRing()
DelayProgram()
Sound.PlayBellRingAndWait()
DelayProgram()
Sound.PlayChime()
DelayProgram()
Sound.PlayChimeAndWait()
DelayProgram()
Sound.PlayChimes()
DelayProgram()
Sound.PlayChimesAndWait()
DelayProgram()
Sound.PlayClick()
DelayProgram()
Sound.PlayClickAndWait()
Sub DelayProgram
  Program.Delay(2000)
EndSub


Stack (Работа со стеком)
Этот объект обеспечивает хранение значений по принципу стопки тарелок. Вы можете поместить значение в стек и извлечь его. Но при этом вы можете извлекать значения только по одному, в порядке их помещения в стек. Последнее помещенное значение будет извлечено первым.

Стек (англ. stack — стопка) — структура данных с методом доступа к элементам LIFO (англ. Last In — First Out, «последним пришел — первым вышел»). Чаще всего принцип работы стека сравнивают со стопкой тарелок: чтобы взять вторую сверху, нужно снять верхнюю.

Добавление элемента, называемое также проталкиванием (push), возможно только в вершину стека (добавленный элемент становится первым сверху). Удаление элемента, называемое также выталкивание (pop), возможно также только из вершины стека, при этом, второй сверху элемент становится верхним.

Стеки широко применяются в вычислительной технике — в частности, для отслеживания точек возврата из подпрограмм используется стек вызовов, который является неотъемлемой частью архитектуры большинства современных процессоров. Языки программирования высокого уровня также используют стек вызовов для передачи параметров при вызове процедур.

Операции
GetCount
Stack.GetCount(stackName) - Получает число элементов стека.

stackName - Имя стека.
Возвращает - Число элементов в указанном стеке.
PopValue
Stack.PopValue(stackName) - Достает значение из указанный стека.

stackName - Имя стека.
Возвращает - Значение из стека.
PushValue
Stack.PushValue(stackName, value) - Добавляет значение в указанный стек.
stackName - Имя стека.
value - Значение, которое надо добавить в стек.
Возвращает – Ничего
	‘Пример программы работы со стеком
Stack.PushValue("первый стек", "первый стек, 1")
Stack.PushValue("первый стек", "первый стек, 2")
Stack.PushValue("первый стек", "первый стек, 3")
Stack.PushValue("второй стек", "второй стек, 1")
Stack.PushValue("второй стек", "второй стек, 2")
Stack.PushValue("второй стек", "второй стек, 3")
Stack.PushValue("второй стек", "второй стек, 4")
TextWindow.WriteLine("Первый стек содержит " + Stack.GetCount("первый стек") + " элемента(ов).")
TextWindow.WriteLine("Второй стек содержит " + Stack.GetCount("второй стек") + " элемента(ов).")
TextWindow.WriteLine("Верхний элемент для первого стека это: " + Stack.PopValue("первый стек") + ".")
TextWindow.WriteLine("Верхний элемент для второго стека это: " + Stack.PopValue("второй стек") + ".")
TextWindow.WriteLine("После извлечения, верхний элемент первого стека это: " + Stack.PopValue("первый стек") + ".")
TextWindow.WriteLine("После извлечения, верхний элемент первого стека это: " + Stack.PopValue("второй стек") + ".")


Text (Работа с текстом)
Объект Text обеспечивает работу с текстом.
Операции
Append
Text.Append(text1, text2) - Соединяет два текстовых значения и возвращает полученное значение в качестве результата. Эта операция особенно полезна, когда осуществляется обработка неизвестного текста, который случайно может быть воспринят программой как число, чтобы избежать суммирования чисел.
text1 - Первая часть текста для добавления.
text2 - Вторая часть текста для добавления.
Возвращает - Полученный текст, содержащий обе указанные части.
ConvertToLowerCase
Text.ConvertToLowerCase(text) - Конвертирует все символы указанного текста в нижний регистр.

text - Текст для преобразования в нижний регистр.
Возвращает - Текст, преобразованный в нижний регистр.
ConvertToUpperCase
Text.ConvertToUpperCase(text) - Конвертирует все символы указанного текста в верхний регистр.

text - Текст для преобразования в верхний регистр.
Возвращает - Текст, преобразованный в верхний регистр.
EndsWith
Text.EndsWith(text, subText) - Проверяет, заканчивается ли заданный текст указанным текстом.
text - Текст, в котором надо осуществить поиск.
subText - Подстрока для поиска.
Возвращает - Возвращает значение "True", если подстрока найдена в конце заданного текста.
GetCharacter
Text.GetCharacter(characterCode) - Получает символ по его коду Юникод. В дальнейшем символ может быть использован в простом тексте.

characterCode - Код в стандарте Юникод для нужного символа.
Возвращает - Символ, который соответствует заданному коду в стандарте Юникод.
GetCharacterCode
Text.GetCharacterCode(character) - Получает код в стандарте Юникод  для указанного символа.

character - Символ, для которого надо получить код в стандарте Юникод.
Возвращает - Код в стандарте Юникод для указанного символа.
GetIndexOf

Text.GetIndexOf(text, subText) - Находит позицию в тексте, с которой начинается указанная подстрока.

Text - Текст, в котором осуществляется поиск.

subtext - Текст, который надо найти.

Возвращает - Возвращает позицию в тексте, на которой начинается заданная подстрока. Если подстрока не найдена, возвращает -1.

GetLength
Text.GetLength(text) - Получает длину указанного текста.

text - Текст, длину которого надо вычислить.
Возвращает - Размер указанного текста.
GetSubText
Text.GetSubText(text, start, length) - Получает часть указанного текста.
text - Текст, из которого надо извлечь подстроку.
start - Указывает начальную позицию.
length - Задает длину подстроки.
Возвращает - Запрошенная часть текста.
GetSubTextToEnd

Text.GetSubTextToEnd(text, start) - Считывает подстроку из заданного текста, начиная с указанной позиции.

text - Текст, из которого надо извлечь подстроку.

start - Указывает начальную позицию.

Возвращает - Запрошенная подстрока.

IsSubText
Text.IsSubText(text, subText) - Проверяет, является ли указанный текст частью большего текста.
text - Текста, в котором будет искаться подстрока.
subText - Подстрока для поиска.
Возвращает - Возвращает значение "True", если подстрока найдена в заданном тексте.
StartsWith
Text.StartsWith(text, subText) - Проверяет, начинается ли заданный текст определенным текстом.
text - Размер текста для осуществления поиска.
subText - Подстрока для поиска.
Возвращает - Возвращает значение "True", если подстрока найдена в начале заданного текста.
	Вот пример кода демонстрирующий, как использовать все операции класса Text в Microsoft Small Basic:

TextWindow.Write("Как Вас зовут? ")
firstName = TextWindow.Read()
TextWindow.Write("Какая у Вас фамилия? ")
lastName = TextWindow.Read()
firstNameWithSpace = Text.Append(firstName, " ")
fullName = Text.Append(firstNameWithSpace, lastName)
TextWindow.WriteLine("Ваше полное имя " + fullName + ".")
TextWindow.WriteLine("Версия все буквы строчные - " + Text.ConvertToLowerCase(fullName) + ".")
TextWindow.WriteLine("Версия все буквы заглавные - " + Text.ConvertToUpperCase(fullName) + ".")
TextWindow.WriteLine("Ваше полное имя заканчивается символом 'e': " + Text.EndsWith(fullName, "e") + ".")
TextWindow.WriteLine("Позиция первого символа `a` в Вашем полном имени: " + Text.GetIndexOf(fullName, "a") + ".")
TextWindow.WriteLine("Ваше полное имя содержит - " + (Text.GetLength(fullName) - 1) + " знаков.")
TextWindow.WriteLine("Первые четыре символа: " + Text.GetSubText(fullName, 1, 4) + ".")
TextWindow.WriteLine("Последние четыре символа: " + Text.GetSubTextToEnd(fullName, Text.GetLength(fullName) - 3) + ".")
TextWindow.WriteLine("Ваше полное имя содержит символ 'a': " + Text.IsSubText(fullName, "a") + ".")
TextWindow.WriteLine("Ваше полное имя начинается с буквы 'P': " + Text.StartsWith(fullName, "P") + ".")
TextWindow.WriteLine("Символ для кода 97: " + Text.GetCharacter(97) + ".")
TextWindow.WriteLine("Код для символа 'a' это: " + Text.GetCharacterCode("a") + ".")

Обратите внимание, что есть операции для того, чтобы добавить текст, печатать его строчными буквами или прописными буквами, а также несколько операций для сравнения и извлечения текста. Есть также несколько операций для того, чтобы конвертировать текст между символами и символьными кодами.


TextWindow ( Работа с текстовым окном)
Объект TextWindow обеспечивает ввод-вывод информации в режиме текстового окна. Например, используя этот класс можно вывести на экран текст или число или получить данные из текстового окна.
Свойства
BackgroundColor
TextWindow.BackgroundColor - Получает или устанавливает цвет фона для текста в текстовом окне.
CursorLeft
TextWindow.CursorLeft - Получает или устанавливает колонку позиции курсора в текстовом окне.
CursorTop
TextWindow.CursorTop - Получает или устанавливает строку позиции курсора в текстовом окне.
ForegroundColor
TextWindow.ForegroundColor - Получает или устанавливает цвет текста в текстовом окне.
Left
TextWindow.Left - Получает или устанавливает позицию левой границы Текстового Окна.
Title
TextWindow.Title - Получает или устанавливает заголовок текстового окна.
Top
TextWindow.Top - Получает или устанавливает позицию верхней границы Текстового Окна.
Операции
Clear
TextWindow.Clear() - Очищает текстовое окно.
Возвращает - Ничего
Hide
TextWindow.Hide() - Скрывает текстовое окно.
Возвращает - Ничего
Pause
TextWindow.Pause() - Ожидает ввода данных пользователем.
Возвращает - Ничего
PauseIfVisible
TextWindow.PauseIfVisible() - Ожидает ввода данных пользователем, только если текстовое окно уже открыто.
Возвращает – Ничего
PauseWithoutMessage
TextWindow.PauseWithoutMessage() - Ожидает ввода данных пользователем.
Возвращает - Ничего
Read
TextWindow.Read() - Считывает строку текста из текстового окна. Функци не возвращает результат, пока не нажата клавиша Ввод (Enter).
Возвращает -  Текст, который был считан из текстового окна.
ReadNumber
TextWindow.ReadNumber() - Считывает числовое значение из текстового окна. Функци не возвращает результат, пока не нажата клавиша Ввод (Enter).
Возвращает - Число, которое было считано из текстового окна.
Show
TextWindow.Show() - Показывает текстовое окно.
Возвращает - Ничего
Write
TextWindow.Write(data) - Выводит текст или число в текстовое окно. В отличие от метода WriteLine, последующий текст будет появляться на той же строке.

data - Текст или число, которое надо записать в текстовое окно.
Возвращает - Ничего
WriteLine
TextWindow.WriteLine(data) - Выводит текст или число в текстовое окно. К концу строки будет добавлен символ переноса строки, так что последующий текст будет выводиться с новой строки.

data - Текст или число, которое надо вывести в текстовом окне.
Возвращает – Ничего
	Это один из первых примеров, которые демонстрируют, как использовать Microsoft, Small Basic v. 0.8 для программирования на компьютере. Этот пример предполагает, что Вы уже загрузили и установили Small Basic и что Вы уже прочли, по крайней мере, первые 10 страниц «Введения в Small Basic». Мы начнем с обсуждения возможностей текстового окна, которое обеспечивает связанные с текстом функциональные возможности ввода и вывода. Например, Вы можете использовать текстовое окно, чтобы получить некоторый текст и числа от пользователей и также написать некоторый текст и числа на экране компьютера.
Попытайтесь набрать или скопировать этот код: в пустое окно редактора Small Basic и затем нажмите клавишу F5 (или, на панели инструментов, нажмите Run (F5)).
' Показывает и очищает содержимое текстового окна.
' Команды отображения и очистки текстового окна здесь не нужны, они включены для демонстрации.
TextWindow.Show()
TextWindow.Clear()
' Установка цвета выделения для текста в желтый и изменение цвета текста на синий в текстовом окне.
TextWindow.BackgroundColor = "Yellow"
TextWindow.ForegroundColor = "Blue"
' Устанавливает курсор в положение 15 столбцов слева и 10 строк вниз от верхнего левого угла текстового окна.
TextWindow.CursorLeft = 15
TextWindow.CursorTop = 10
' Устанавливает текстовое окнов положение 200 пикселей вправо и 400 пикселей вниз от левого верхнего угла экрана.
TextWindow.Left = 200
TextWindow.Top = 400
' Устанавливает заголовок для текстового окна.
TextWindow.Title = "Это - мой заголовок!"
' Получение информации от пользователя и затем отображение ее назад пользователю.
TextWindow.WriteLine("Введите любой текст:")
myText = TextWindow.Read()
TextWindow.WriteLine("Введите любое число:")
myNumber = TextWindow.ReadNumber()
TextWindow.Write("Вы ввели " + myText + " и " + myNumber + ". ")
' Ожидание действия пользователя, чтобы затем скрыть текстовое окно.
' Команды паузы и скрываня текстового окна здесь не нужны, они включены для демонстрации.
TextWindow.Pause()
TextWindow.Hide()
Здесь Вы можете обратить внимание на несколько вещей которые могут показаться не понятными:

· BackgroundColor является свойством, которое является характеристикой или атрибутом текстового окнф. " Yellow " и "Blue", являются собственно значениями BackgroundColor – означающие название используемого цвета. Есть и другие допустимые значения, которые можно использовать здесь, например, "Green" и "Magenta" и т.д. Описание всех допустимых значений цветов приводится в «Введении в Small Basic»
· myText и MyNumber являются переменными. Вы можете использовать переменные для временного хранения значений, а затем использовать их позже в своей программе.

· Круглые скобки используются, чтобы передать параметр (константу, переменную или выражение) или специальную информацию операцией WriteLine в объект TextWindow. Объект может что-то выполнить, а параметр в операции определяет то, что может (нужно) выполнить объекту. Обратите внимание, что пустые круглые скобки означают, что нет необходимости передавать никаких особых параметров с этой операцией в объект.


Timer (Работа с повтором действий)
Объект Timer обеспечивает способ повторения действий через постоянные интервалы времени.
Свойства
Interval

Timer.Interval - Возвращает или устанавливает интервал срабатывания таймера в миллисекундах. Может принимать значения от 10 до 100000000.
Операции

Pause

Timer.Pause() - Приостанавливает таймер. События Tick таймера не вызываются.

Возвращает - Ничего

Resume

Timer.Resume() - Возобновляет работу таймера после приостановки. События Tick таймера начинают вызываться.

Возвращает - Ничего

События
Tick
Timer.Tick - Вызывает событие Tick для таймера.
Turtle (Работа с черепашкой)
Черепашка (Turtle) позволяет рисовать объекты путем изменения свойств пера (Pen) и рисования примитивных фигур, наподобие языка Logo.
Свойства
Angle
Turtle.Angle - Получает или задает текущий угол движения черепашки. Черепашка поворачивается на заданный угол мгновенно.
Speed
Turtle.Speed - Указывает как быстро должна перемещаться черепашка. Разрешенные значения от 1 до 10. Если выставлена скорость 10, то черепашка перемещается мгновенно.
X
Turtle.X - Получает или задает Х координату черепашки. Черепашка перемещается на новое место мгновенно.
Y
Turtle.Y - Получает или задает Y координату черепашки. Черепашка перемещается на новое место мгновенно.
Операции
Hide
Turtle.Hide() - Прячет черепашку и запрещает операции с ней.
Возвращает - Ничего
Move
Turtle.Move(distance) - Перемещает черепашку на заданное расстояние.  Если перо включено, будет нарисована линия по пути движения черепашки.

distance - Расстояние, на которое должна переместиться черепашка.
Возвращает - Ничего
MoveTo

Turtle.MoveTo(x, y) - Поворачивает и перемещает черепашку в указанное расположение. Если перо включено, то будет нарисована линия по пути движения черепашки.

X - Координата X конечной точки.

Y - Координата Y конечной точки.

Возвращает -  Ничего
PenDown
Turtle.PenDown() - Включает перо, повторяющее движения черепашки.
Возвращает - Ничего
PenUp
Turtle.PenUp() - Отключает перо, повторяющее движения черепашки.
Возвращает - Ничего
Show
Turtle.Show() - Показывает черепашку и разрешает операции с ней.
Возвращает - Ничего
Turn
Turtle.Turn(angle) - Поворачивает черепашку на указанный угол. Угол указывается в градусах. Если угол задан положительной величиной, то черепашка повернет направо. Если отрицательной - налево.

angle - Угол поворота черепашки.
Возвращает - Ничего
TurnLeft
Turtle.TurnLeft() - Поворачивает черепашку на 90 градусов влево.
Возвращает - Ничего
TurnRight
Turtle.TurnRight() - Поворачивает черепашку на 90 градусов вправо.
Возвращает – Ничего
	Microsoft Small Basic предоставляет класс Turtle (Черепашка), который позволяет Вам привлекать в графическое окно небольшую черепашку. Вы можете сделать ползание черепашки в графическом окне и рисуя везде, где только Вы её проводите, как демонстрируется в следующем коде, который рисует слово «Hi!» («Привет!») около середины графического окна:
' Центр графического окна.
w = GraphicsWindow.Width / 2
h = GraphicsWindow.Height / 2
' Переместите черепаху в центр графического окна.
Turtle.Show()
Turtle.X = w
Turtle.Y = h
' Заставьте черепаху рисовать быстро (но не слишком быстро!).
Turtle.Speed = 7
' Нарисуйте “H” в "Hi!"
Turtle.Turn(180) ' Поворачивает черепаху на указанный угол с указанной скоростью.
Turtle.Move(40)
Turtle.MoveTo(w, h + 20) ' То же самое что и координаты (X и Y) черепахи, но перемещение на указанной скорости вместо немедленно.
Turtle.TurnRight()
Turtle.Move(20)
Turtle.MoveTo(w + 20, h) 

Turtle.Angle = 180 ' То же самое как Turtle.Turn(180), но черепахаповорачивается на заданный угол немедленно.
Turtle.Move(40)
' Нарисуйте "i" в "Hi!"
Turtle.X = w + 30 ' То же самое как первая часть Turtle.MoveTo, но перемещается немедленно и не рисует, двигаясь.
Turtle.Y = h ' То же самое как вторая часть Turtle.MoveTo, но перемещается немедленно и не рисует, двигаясь.
Turtle.PenUp() ' Теперь, когда черепаха двигается, она не будет рисовать, так как перо поднято.
Turtle.Move(10)
Turtle.PenDown() ' Теперь черепаха рисует, поскольку она двигается и перо опущено.
Turtle.Move(5)
Turtle.PenUp() ' Остановка рисования. Подымание пера.
Turtle.Move(5)
Turtle.PenDown() ' Начало рисования. Опускание пера.
Turtle.Move(20)
' Нарисуйте "!" в "Hi!"
Turtle.X = w + 40
Turtle.Y = h
Turtle.Move(30)
Turtle.PenUp()
Turtle.Move(5)
Turtle.PenDown()
Turtle.Move(5)
Turtle.Hide()
Хотя не совсем как развлечение, Вы можете выполнить ту же самую надпись с помощью операции DrawLine в графическом окне, как демонстрируется в следующем коде:
' Центр графического окна.
w = GraphicsWindow.Width / 2
h = GraphicsWindow.Height / 2
' Нарисуйте "H" в "Hi!"
GraphicsWindow.DrawLine(w, h, w, h + 40)
GraphicsWindow.DrawLine(w, h + 20, w + 20, h + 20)
GraphicsWindow.DrawLine(w + 20, h, w + 20, h + 40)
' Нарисуйте "i" в "Hi!"
GraphicsWindow.DrawLine(w + 30, h + 10, w + 30, h + 15)
GraphicsWindow.DrawLine(w + 30, h + 20, w + 30, h + 40)
' Нарисуйте "!" в "Hi!"
GraphicsWindow.DrawLine(w + 40, h, w + 40, h + 30)
GraphicsWindow.DrawLine(w + 40, h + 35, w + 40, h + 40)


Ключевые слова
For
Оператор For обеспечивает многократное выполнение набора операторов.

Пример - Данный пример выводит на экран числа от 1 до 10:
For i = 1 To 10
TextWindow.WriteLine(i)
EndFor

Goto
Оператор Goto позволяет переходить в любую точку программы.

Пример - Данный пример безостановочно выводит на экран последовательные числа:

start:
TextWindow.WriteLine(i)
i = i + 1
Goto start
If
Оператор If позволяет задать условие, при котором выполняются различные действия.

Пример - Данный пример выводит на экран в зависимости от выпавшего жребия "Выигрыш" или "Проигрыш":

If flip = "Решка" Then
TextWindow.WriteLine("Выигрыш")
Else
TextWindow.WriteLine("Проигрыш")
EndIf

Sub
Оператор Sub (Subroutine - процедура) позволяет группировать ряд действий для выполнения их с помощью одного вызова процедуры.

Пример - Данный пример содержит процедуру, которая сообщает о победе и выводит на экран "Выигрыш":

Sub Win
Sound.PlayBellRing()
TextWindow.WriteLine("Выигрыш!")
EndSub

While
Оператор While обеспечивает выполнение действий до тех пор, пока не будет достигнут требуемый результат.

Пример - Данный пример выводит на экран набор случайных чисел до тех пор, пока не выпадет число больше 100.
While i < 100
i = Math.GetRandomNumber()
TextWindow.WriteLine(i)
EndWhile
	' Примеры работы с ключевыми словами

' Весь диапазона от 1 до 10.
' Печатать цифры от 1 до 10.
For i = 1 To 10 

  TextWindow.Write(i + " ")
EndFor
TextWindow.WriteLine("")
' Использование заданного шага
' Печатать только - 1, 3, 5, 7, 9.
For i = 1 To 10 Step 2
  TextWindow.Write(i + " ")
EndFor
TextWindow.WriteLine("")
' Использование специального условия While (Пока).
' Печатать только от 1 до 9.
i = 1
While i < 10
  TextWindow.Write(i + " ")
  i = i + 1  

EndWhile
TextWindow.WriteLine("")
' Использование специального условия If (Если).
If Clock.Year = 2010 Then
  TextWindow.WriteLine("Год - 2010.")
EndIf
' Использование специального условия If/Else (Если/Иначе).
If Clock.Year = 2010 Then
  TextWindow.WriteLine("Год - 2010.")
Else
  TextWindow.WriteLine("Год не 2010.")
EndIf
' Использование специального условия If/ElseIf (Если/ИначеЕсли).
If Clock.Year = 2010 Then
  TextWindow.WriteLine("Год - 2010.")
ElseIf Clock.Year = 2009 then
  TextWindow.WriteLine("Год - 2009.")
Else
  TextWindow.WriteLine("Год не 2009 и не 2010.")
EndIf
Вы можете также управлять выполнение программы, используя следующие ключевые слова GoTo, Sub и EndSub, как демонстрируется ниже:
Sub PrintSomeMoreText
  TextWindow.WriteLine("Четвертая строка для вывода.")  

EndSub
' Начало программы.
TextWindow.WriteLine("Первая строка для вывода.")
Sub PrintSomeText
  TextWindow.WriteLine("Третья строка для вывода.")
EndSub
TextWindow.WriteLine("Вторая строка для вывода.")
PrintSomeText() ' Обращение к процедуре.
Goto PrintMoreText ' Безусловный переход к процедуре.
goodbye:
TextWindow.WriteLine("До встречи!")
Goto end
Sub ThisWillNotPrint
  TextWindow.WriteLine("Это не будет напечатано, если я не вызову это из программы.")
EndSub
printMoreText:
PrintSomeMoreText() ' Обращение к процедуре.
Goto goodbye ' Безусловный переход к метке.
end:

